
 *------------------------------------------------* 

// Modified non-differential simulation program // 

*------------------------------------------------* 

 

* Author: Peter J Godolphin 

* Version: 2 

* Date created: 14 March 2019 

* Last modified: 03 January 2020 

* Programmed in: Stata 15.1 

 

/* DESCRIPTION: 

This program creates a dataset with a binary outcome. The parameters used to  

create this dataset can be modified to mimic specific trial circumstances. The  

program then misclassifies a random selection of the participants to ensure that 

the treatment effect remains constant. The outcome is analysed to produce a risk 

ratio and this output is produced, as is the number of misclassifications. */ 

 

/* PARAMETERS: 

- N: Sample size (set up for 3000) 

- A: Event rate (set up for 10%) 

- B: Proportion of events in control group (set up for 60%) 

- C: Proportion of events in treatment group (set up for 40%) 

// B & C are used to determine treatment effect, here it is 0.67 

// Setting B to 0.55 and C to 0.45 will give 0.82 

// Setting B to 0.525 and C to 0.475 will give 0.90 

- X: Sets level of non-differential  misclassification 

// Users can increase X until the treatment effect is non significant */   

 

*---------* 



// START // 

*---------* 

 

capture program drop simulate_non 

 program define simulate_non  

 version 15.1 

 syntax, [N(int 3000) A(real 0.1) B(real 0.6) C(real 0.4) X(real 0)] 

  

 * Set up dataset with desired event rate and true treatment effect 

 clear 

 set obs `n' 

 qui gen refno = _n 

 qui gen trt = 1 if refno <= `n'/2 

 qui replace trt = 2 if refno > `n'/2 

 qui gen outcome = 1 if refno <= `n'*`a'*`b' 

 qui replace outcome = 1 if refno > `n'/2 & refno <= (`n'/2 + `n'*`a'*`c') 

 qui replace outcome = 0 if outcome == . 

  

 * Non-differential misclassification  

 set seed 1111 

 qui gen rand1 = runiform() if trt == 1 & outcome == 1 

 sort rand1 

 qui gen n1 = _n if trt == 1 & outcome == 1 

 qui replace outcome = 0 if n1 <= `b'*`x'*10 

 qui gen count1 = 1 if n1 <= `b'*`x'*10 

 qui gen rand2 = runiform() if trt == 2 & outcome == 1 

 sort rand2 

 qui gen n2 = _n if trt == 2 & outcome == 1 

 qui replace outcome = 0 if n2 <= `c'*`x'*10 



 qui gen count2 = 1 if n2 <= `c'*`x'*10 

 qui sum count1 

 gen c1 = `r(N)' 

 qui sum count2 

 qui gen c2 = `r(N)' 

 qui gen count = c1 + c2 

   

 glm outcome trt, family(binomial) link(log) eform 

  

 * Display the number of misclassifications 

 qui sum count 

 display as text "Total number of misclassifications: `r(mean)'" 

end 

 

*-------* 

// END // 

*-------* 


